

Making CET quantities add up

Mark Horridge, September 2014

1. Introduction

The CES functional form is heavily used in CGE modelling to combine several inputs into one aggregate. Cousin to CES is the less familiar CET form, which is used to split one thing into several. For example, CET might be used to

- split total output of some good into output for consumption and output for export.
- allocate a fixed stock of labour between industries.
- allocate a fixed stock of land between agricultural industries.

Below we focus on the last, land, example.

The CET equations often have the percentage change form:

$$x_i = x_{tot} + \sigma(p_i - p_{ave})$$

$$p_{ave} = \sum_i S_i \cdot p_i \quad \text{where } S_i \text{ is the share of industry } i \text{ in total land rentals.}$$

which is the same as CES except for the positive sign on σ , the *Constant Elasticity of Transformation*. However, where high or infinite values of σ are regarded as plausible; it is necessary to rewrite the CET equations as:

$$(p_i - p_{ave}) = \tau(x_i - x_{tot}) \quad \text{where } \tau = 1/\sigma \quad [\tau=0 \rightarrow \sigma=\infty]$$

$$x_{tot} = \sum_i S_i \cdot x_i$$

A simple TABLO implementation of the CET appears below:

CET1.TAB: a simple CET implementation

```
File INFILE;
! Data from file !
Set CROPS # Alternate land uses # read elements from file INFILE header "CROP";
Coefficient (all,c,CROPS) RENT(c) # Crop revenues #;
Read RENT from file INFILE header "RENT";
Coefficient (parameter) SIGMA # CET elasticity #;
Read SIGMA from file INFILE header "SIG";
! Derived coefficients !
Coefficient
TOTRENT # Total crop revenue #;
(all,c,CROPS) RENTSHR(c) # Revenue shares #;
Formula
TOTRENT = sum{c,CROPS, RENT(c)};
(all,c,CROPS) RENTSHR(c) = RENT(c)/TOTRENT;
Variable
(all,c,CROPS) p(c) # Rent per effective land units #;
(all,c,CROPS) x(c) # Quantities of effective land units #;
xtot # Total output (revenue-weighted) #;
pave # Average rent (revenue-weighted) #;
Update (all,c,CROPS) RENT(c) = p(c)*x(c);
Equation
E_pave pave = sum{c,CROPS, RENTSHR(c)*p(c)};
E_x (all,c,CROPS) x(c) = xtot + SIGMA*(p(c) - pave);
```

which is driven by the following CMF file¹:

CET1.CMF: simulation for CET1.TAB

```

auxiliary files = cet1;
File INFILE = DATA.HAR;
Updated File INFILE = <cmf>.UPD;
log file = yes;
method = Gragg ;
steps = 10 20 30;
Verbal Description = Wheat/Beef price changes;
Shock p("Wheat") = 20;
Shock p("Beef") = -20;
Exogenous p xtot;
Rest endogenous;

```

Data and results from CET1 appear in Table 1 below; results for p_{ave} and x_{tot} are shown in the Total row of the two final columns.

Table 1: Data and results from CET1.CMF

	Rent	RentShr	p	x
Wheat	5000	0.50	20.00	49.33
Fruit	3000	0.30	0.00	-39.99
Beef	2000	0.20	-20.00	-80.33
Total	10000	1.00	10.75	0.00

In DATA.HAR, the value of σ is 5. As we would expect, the output mix shifts strongly towards Wheat and away from Fruit and especially Beef.

So far, so good. The next step is to add to CET1.TAB some data for land areas (in hectares). As a check, we compute q_{tot} , the change in total land area (we hope there is no change). Existing results are unaffected.

Additions for CET2.TAB: adding land areas

```

! As for CET1.TAB, plus .... !
Coefficient (all,c,CROPS) AREA(c) # Hectares used by crops #;
Read AREA from file INFILE header "AREA";
Coefficient
  TOTAREA # Total crop area #;
  (all,c,CROPS) AREASHR(c) # Area shares #;
Formula
  TOTAREA = sum{c,CROPS, AREA(c)};
  (all,c,CROPS) AREASHR(c) = AREA(c)/TOTAREA;
Update (all,c,CROPS) AREA(c) = x(c);
Variable qtot # Total land area #;
Equation E_qtot qtot = sum{c,CROPS, AREASHR(c)*x(c)};

```

Table 2: Data and simple CET results with land areas

	Rent	Area	RentShr	AreaShr	p	x	q
Wheat	5000	200	0.50	0.20	20.00	49.33	49.33
Fruit	3000	300	0.30	0.30	0.00	-39.99	-39.99
Beef	2000	500	0.20	0.50	-20.00	-80.33	-80.33
Total	10000	1000	1.00	1.00	10.75	0.00	-42.30

¹ Accompanying this document are all TAB, CMF, and HAR files needed to reproduce the simulations.

Above, we assume that $q(i)$, the %change in land area, follows $x(i)$, the "effective" land quantities, $x(i)$. Results for $qtot$ appear in the bottom right cell. We see quite a large change in total land area! This tells us that the $x(i)=q(i)$ cannot be used to report changes in land areas. The discrepancy between $xtot$ and $qtot$ is inevitable given the large differences between the $RentShr$ and the $AreaShr$ vectors. Such discrepancies might arise from 2 causes:

- Quite likely the $Rent$ and $Area$ values are drawn from different data sources which employ different conventions. For example, there may be differences in sectoral definitions. Perhaps more care with the data would reduce the problem.
- Even if the data are correct, the $RentShr$ and $AreaShr$ will diverge because average unit rent per hectare will differ between uses. In this example, the average unit rent per hectare is much lower for beef -- which is entirely plausible. We need to realize that for each crop there is a marginal rent per hectare which will differ from the average rent. At the margin of substitution between Beef land and Wheat land profits per hectare (marginal area rents) must be equal. That does not mean that average rents must be equal².

More generally the problem arises because the natural unit of measurement in CGE models is base-period-dollars-worth (the amount that initially may be bought with \$1). As soon as we introduce another unit of measurement (eg, hectares) problems may arise.

Although the units problem often appears in a CET or supply-side context, it may also appear in a CES or demand-side context. For example, an electricity distributor may purchase electricity from solar, coal or nuclear generator -- regarding these sources as imperfect substitutes. Average costs per kilowatt-hour will vary between sources -- implying that $qtot$ and $xtot$ measures of aggregate use will differ. Since, for the final user, a coal kilowatt-hour is indistinguishable from a solar kilowatt-hour, the difference is annoying to explain.

Again, results from a simulation where employment is fixed will depend on whether the fixed employment total is an hours-weighted or wage-weighted aggregate. Usually the latter is chosen, -- implying that one new employed dentist can compensate for 10 lost cleaning jobs.

2. What if $qtot$ were fixed?

A simple remedy might be to run `CET2.TAB` with $qtot$ fixed (instead of $xtot$, see `CET2A.CMF`). With this closure the additional `TABLO` code for `CET2.TAB` *does* affect all results. The $x(i)$ results are greatly changed. Indeed $qtot$, the total land area, is fixed; but $xtot$, the total economic contribution of land, varies quite a lot.

Table 2a: Data and results with fixed total land area

	RentShr	AreaShr	p	x	q
Wheat	0.50	0.20	20.00	158.79	158.79
Fruit	0.30	0.30	0.00	4.00	4.00
Beef	0.20	0.50	-20.00	-65.92	-65.92
Total	1.00	1.00	10.75	73.30	0.00

The large change in $xtot$ raises severe difficulties with the welfare-oriented interpretation of results that is common in policy analysis. It implies that small changes in land allocation can affect GDP and other macro aggregates -- contrary to normal economic intuition. With $qtot$ fixed, a hectare of beef land moving to wheat use enjoys an immediate sharp rise in per-hectare rents, causing

² Unfortunately the CET specification implies that the ratio of average to marginal rents is the same for all crops (see 1st page of Appendix 3) -- which is part of the problem. That is not true for all other functional forms.

aggregate output to rise. This is equivalent to treating the initial database as including a large distortion (good land wasted on the low-value Beef use). Hence, second-best effects will colour policy conclusions -- any policy is good which favours Wheat over Beef.

2.1. A related approach

The approach just described (qtot fixed) has no optimizing interpretation. The following idea addresses this problem. We assume that one land owner distributes land between uses according to the rule:

$$\text{Choose } X_i \text{ to maximize } U = \sum_i [P_i X_i]^\alpha \text{ such that } \sum_i X_i = Q_{\text{tot}}$$

[again we equate X_i with Q_i], giving rise to the following %change FOC:

$$x_i = q_{\text{tot}} + \sigma(p_i - p_{\text{ave}})$$

$$q_{\text{tot}} = \sum_i H_i \cdot x_i \quad \text{where } H_i \text{ is the share of industry } i \text{ in total land area.}$$

Although ingenious, these FOC in fact generate the *same* results as CET2A.CMF.³ Hence we include no TAB file for this approach. All the criticisms of CET2A still apply. The non-optimal behaviour which was implicit in CET2A is here made explicit: the land owner does not value a dollar earned from Wheat as highly as a dollar earned from Beef: livestock (or at least variety) has special, non-monetary, attraction.

3. Methods of scaling land areas to add up

The basic framework of CET1.TAB uses data and variables that measure land in economic units [base-period-dollars-worth]. To tackle the problems described above, modellers have usually added to that framework a parallel system of data and variables which measures land in physical (hectare) units, as shown below.

Table 3: A parallel system of data and variables in physical units

	Economic units	Physical Units
Data	RENT(i), TOTRENT	AREA(i), TOTAREA
Shares	RENTSHR(i)	AREASHR(i)
Quantity variables	x(i), xtot	q(i), qtot
Price variables	p(i), ptot	r(i), rave

The added system enables reporting of results in physical units that add up correctly, yet it does not affect any of the results generated by CET1.TAB. Hence, substitution at the margin has no disturbing welfare effects.

It will be obvious from the preceding discussion that we cannot assume (as we did in CET2.TAB) that the percentage changes $x(i)$ and $q(i)$ are the same. For, if $x(i)=q(i)$, and $AREASHR(i) \neq RENTSHR(i)$, then $q_{\text{tot}} \neq x_{\text{tot}}$ (so they cannot both = 0, as usually desired).

Instead modellers assume, in levels, that

$$Q_i = F(X_i)$$

The choice of functional form F differs between practitioners (and is usually ad hoc).

³ Except that now $p_{\text{ave}} = \sum_i H_i \cdot p_i$

3.1. Method 1

A common choice for F is:

$$Q_i = \Lambda X_i \quad \text{where } \Lambda \text{ is chosen so that } \sum_i Q_i = Q_{\text{tot}} \quad \text{with } Q_{\text{tot}} \text{ exogenous}$$

or, in percent change form:

$$q_i = x_i + \lambda \quad \text{and} \quad \sum_i H_i q_i = q_{\text{tot}} \quad \text{where the } H_i \text{ are hectare shares.}$$

Such a system is shown in CET3.TAB below:

Additions for CET3.TAB: scaling land areas to add up, method 1

```
! As for CET1.TAB, plus .... !
Coefficient (all,c,CROPS) AREA(c) # Hectares used by crops #;
Read AREA from file INFILE header "AREA";
Coefficient
  TOTAREA # Total crop area #;
  (all,c,CROPS) AREASHR(c) # Area shares #;
Formula
  TOTAREA = sum{c,CROPS, AREA(c)};
  (all,c,CROPS) AREASHR(c) = AREA(c)/TOTAREA;
Variable (all,c,CROPS) q(c) # Percent change land areas #;
lambda # slack variable to allow correct land area addup #;
Update (all,c,CROPS) AREA(c) = q(c);
Equation E_q (all,c,CROPS) q(c) = x(c) + lambda;
Variable qtot # Total land area #;
Equation E_qtot qtot = sum{c,CROPS, AREASHR(c)*q(c)};
```

Results for the physical unit price variables, r_i and r_{ave} are not usually computed, but would be given by percent change equations⁴:

$$r_i + q_i = p_i + x_i \quad \text{and} \quad r_{\text{ave}} = \sum_i S_i r_i \quad \text{where the } S_i \text{ are revenue shares.}$$

New results using CET3.TAB are shown in the penultimate, q3, column of the table below. Note that results for p and x are the same as the original CET1 results. As desired, q_{tot} , the area-weighted average of the q_i , is 0. However, the q_i diverge quite widely from the x_i . In one case (Fruit) the sign is different! Should we report that land used for Fruit went up or down?

Table 4: Data and CET results with adjusted land areas

	RentShr	AreaShr	QRatio	p	x	q3	q4
Wheat	0.50	0.20	2.50	20.00	49.33	158.79	128.76
Fruit	0.30	0.30	1.00	0.00	-39.99	4.00	-38.18
Beef	0.20	0.50	0.40	-20.00	-80.33	-65.92	-28.60
Total	1.00	1.00	1.00	10.75	0.00	0.00	0.00

3.2. Method 2

Another choice for F is:

$$Q_i = X_i^{\alpha_i} \quad \text{where } \alpha_i = S_i/H_i \quad [= \text{RENTSHR}(i)/\text{AREASHR}(i)]$$

or, in percent change form:

$$q_i = \alpha_i x_i$$

⁴ These equations are in the supplied TAB files, but are not shown in the TAB excerpts presented in the text.

Easy algebra will confirm that with this system we automatically obtain:

$$\sum_i S_i x_i = x_{\text{tot}} = \sum_i H_i q_i = q_{\text{tot}} \quad \text{ie, } q_{\text{tot}} \text{ need not be exogenous.}$$

Furthermore, x_i and q_i will always have the same sign.

Such a system is shown in CET4.TAB below:

Additions for CET4.TAB: scaling land areas to add up, method 2

```
! As for CET1.TAB, plus .... !
Coefficient (all,c,CROPS) AREA(c) # Hectares used by crops #;
Read AREA from file INFILE header "AREA";
Coefficient
  TOTAREA # Total crop area #;
  (all,c,CROPS) AREASHR(c) # Area shares #;
Formula
  TOTAREA = sum{c,CROPS, AREA(c)};
  (all,c,CROPS) AREASHR(c) = AREA(c)/TOTAREA;
Coefficient (all,c,CROPS) QRATIO(c) # QRATIO=RENTSHR/AREASHR #;
Formula (all,c,CROPS) QRATIO(c) = RENTSHR(c)/AREASHR(c);

Variable (all,c,CROPS) q(c) # Percent change land areas #;
Update (all,c,CROPS) AREA(c) = q(c);
Equation E_q (all,c,CROPS) q(c) = QRATIO(c)*x(c);
Variable qtot # Total land area #;
Equation E_qtot qtot = sum{c,CROPS, AREASHR(c)*q(c)};
```

New results using CET4.TAB are shown in the last, q4, column of Table 4 above. As desired, q_{tot} , the area-weighted average of the q_i , is 0. It is not easy to say which of the q3 or q4 columns most nearly resemble the common x column. However, at least the q4 entries all have the same sign as their x counterparts.

Some theoretical support for the QRATIO formula is given by the material in Appendix 3, which proposes that a CRETH-like functional form may also lead to equations that allow quantities to add up properly, in both economic and quantity units.

4. Conclusion

We reviewed the CET functional form, and explained the difficulty that may arise in getting physical quantities to add up correctly. Several possible solutions were reviewed -- none were completely satisfactory. More research is needed !

Appendix 1: Files distributed with this document

A zip archive accompanies this document; it contains:

- TAB and CMF files so you can run all the examples mentioned in the text. Type "RunSims.bat" from the command line to run all the CET examples.
- A complete set of SL4 solution files which you can examine even if you do not want to run all the simulations.

Appendix 2: Special values of σ with CES/CET % change equations

CES equations often have the percentage change form:

$$x_i - x_{tot} = -\sigma(p_i - p_{ave})$$

$$p_{ave} = \sum_i S_i \cdot p_i \quad \text{where the } S_i \text{ are cost shares}$$

CET is the same as CES, except it has a positive sign on σ .

Here we ask the question, what values of σ are admissible?

If σ is very high, the above equations reduce to:

$$p_i = p_{ave} \quad \text{and} \quad p_{ave} = \sum_i S_i \cdot p_i$$

The second equation above is implied by the first, so this system would give rise to a "singular matrix" error. As pointed out above, the following form works better for high values of σ .

$$(p_i - p_{ave}) = -\tau(x_i - x_{tot}) \quad \text{where } \tau=1/\sigma \quad [\tau=0 \rightarrow \sigma=\infty]$$

$$x_{tot} = \sum_i S_i \cdot x_i$$

but would not work for near-zero values of σ . However, both forms work fine when $\sigma=1$ (levels forms generally have a problem with this Cobb-Douglas case).

Generally the percentage change CES equations consist of two equations. First, ONE of the following two:

A $x_i - x_{tot} = -\sigma(p_i - p_{ave})$

B $(p_i - p_{ave}) = -\tau(x_i - x_{tot})$

and second, ONE of the following three:

1 $p_{ave} = \sum_i S_i \cdot p_i$

2 $x_{tot} = \sum_i S_i \cdot x_i$

3 $x_{tot} + p_{ave} = \sum_i S_i \cdot [x_i + p_i]$

The last, 3, is a true zero-profits condition. We have omitted nest-specific technical change terms: if used, they would appear in all the above equations *except* (3) -- a possible advantage of (3).

Permissible combinations of [A,B] and [1,2,3] are:

σ close to 0	σ close to 1	σ very large
A1, A3	A1, A2	A2, A3

You can also write CES as *three* equations:

$$x_i = -\sigma \cdot p_i + \lambda \quad p_{ave} = \sum_i S_i p_i \quad x_{tot} = \sum_i S_i \cdot x_i$$

The extra equation determines λ (which is a combination of non-subscripted terms in the original demand equation).

Appendix 3: A CRETH-like distribution of land between crops

Mark Horridge (Draft: 2008, amended 2014)

[nb: this is a draft, with imperfections -- I hope I can improve it]

A farmer must allocate fixed total land L between several crops (indexed i). Some land suits particular crops, so that crop output, X_i , is a crop-specific function of the land used, L_i . Revenue received is $R = \sum P_i X_i$.

To maximize profit the farmer must choose L_i (or X_i) to maximise revenue $R = \sum P_i X_i$ subject to the constraints $X_i = F(L_i)$ and $\sum L_i = L$.

We observe initial $V_i = P_i X_i$ and the acreages L_i . Revenue and acreage shares are denoted by:

$$S_i = V_i / \sum V_k \quad \text{and} \quad M_i = L_i / \sum L_k$$

Crop yield functions like: $X_i = F(L_i) = A_i L_i^\alpha$ $0 < \alpha < 1$ and $F'(L_i) = [\alpha/L_i] A_i L_i^\alpha = [\alpha/L_i] X_i$ give rise to familiar CET type percent change forms:

$$x_i = x + \sigma[p_i - p] \quad \text{where} \quad x = \sum S_i x_i \quad \text{and} \quad p = \sum S_i p_i$$

and the l_i are given by (using $X_i = A_i L_i^\alpha$):

$$x_i = \alpha l_i \quad \text{where} \quad \alpha \text{ and } \sigma \text{ are related.}$$

We observe however that if the S_i do not equal the M_i , even with L fixed [$\sum L_i l_i = 0$], aggregate real output x changes, giving rise to mysterious efficiency gains. The levels form of the FOC give a clue what is wrong:

$$\text{FOC} \quad P_i F'(L_i) = P_j F'(L_j)$$

or $P_i X_i [\alpha/L_i] = P_j X_j [\alpha/L_j]$ the α cancel out so we get

$$P_i X_i / P_j X_j = L_i / L_j \quad \text{so land and revenue shares must be equal !}$$

In other words, to calibrate this CET-like form, we **require** that $S_i = M_i$ or that initial revenue per acre is equal for each crop. Actual data may not follow this rule.

Intuition: with the crop yield function: $X_i = A_i L_i^\alpha$, the A_i are 'used up' in calibrating the S_i , so there is no flexibility left to accommodate different M_i . However, if the α exponent varied over i , the α_i would not have cancelled out, and our FOC above would become:

$$P_i X_i [\alpha_i/L_i] = P_j X_j [\alpha_j/L_j]$$

$$P_i X_i / P_j X_j = [\alpha_j/\alpha_i] [L_i/L_j]$$

This looks promising and suggests a CRETH form.

To develop the CRETH form, we start with inverse crop yield functions:

$$[L_i/L] = [A_i X_i/X]^{h(i)} \quad \text{where } X \text{ is aggregate real output} \quad \text{note } h_i > 1$$

and the land constraint:

$$\sum [L_i/L] = 1$$

giving the standard CRETH form:

$$\sum [A_i X_i/X]^{h(i)} = 1$$

We get percent change forms:

so $x_i - x = -\sigma_i [p_i - \sum R_k p_k]$ $\sigma_i < 0$ $\sigma_i = 1/[1-h_i]$ $[h_i-1] = -1/\sigma_i$ $h_i = 1 - 1/\sigma_i$ note $h_i > 1$

where the R_k are 'modified' shares given by

$$R_k = \sigma_k S_k / \sum \sigma_j S_j$$

We can add the percent change output proportion equations, weighting by S_i , to get:

$$\sum S_i [x_i - x] = -\sum S_i \sigma_i [p_i - \sum R_k p_k]$$

The RHS vanishes, implying (as we would hope):

$$x = \sum S_i x_i$$

The crop yield functions imply that corresponding land use is given by:

$$[l_i-1] = h_i [x_i - x]$$

or $[x_i - x] = [1/h_i][l_i-1]$

so $[1/h_i][l_i-1] = -\sigma_i [p_i - \sum R_k p_k]$

Again weighting by S_i and adding:

$$\sum S_i [1/h_i][l_i-1] = -\sum S_i \sigma_i [p_i - \sum R_k p_k] = 0$$

or $\sum S_i [1/h_i][l_i-1] = 0$

For all FOC to be satisfied, we require that the above equation is equivalent to:

$$\sum M_i [l_i-1] = 0 \quad \text{percent form of land constraint}$$

that is, the acreage shares M_i must be proportional to $S_i [1/h_i]$

or h_i proportional to S_i/M_i

This fixes relativities within the h_i , but their average level is still 'free'. Our calibration procedure is as follows:

Given initial acreage shares M_i and crop value shares S_i , define ratios:

$$Q_i = S_i/M_i$$

Choose $h > 1$, the desired average of the h_i , and then set

$$h_i = Q_i h$$

Then set: $\sigma_i = 1/[1-h_i]$ [note $h_i > 1$ and $\sigma_i < 0$, so as h_i rises, σ_i tends to 0]

The chosen value of h will determine an 'average' σ , but the dispersion of the σ_i will be controlled by the observed Q_i . Note the CRESH (constant ratio of elasticities) property:

$$AES = \sigma_{ik} = \sigma_i \sigma_k / \sigma \quad \text{where } \sigma \text{ is given by } \sum S_i \sigma_i$$

Crops with higher dollar value per acre will exhibit less volatile output response.

Notes:

A: we require $h_i = Q_i h > 1$ requiring that $h > 1/Q_i$ for all i

In practice we will require that $h > 1.1/Q_i$


meaning that a lower bound for h is the maximum (over i) of the $1.1/Q_i$

$\sigma_i = 1/[1-h_i]$ implies $1-h_i = 1/\sigma_i$ implies $h_i = 1 - 1/\sigma_i$ so if $h=1.1$, $\sigma=-10$

B: As written above, the CRETH system results in small, but not zero change in total area l . Initially we set h_i proportional to S_i/M_i , but the ratios S_i/M_i change as the system is shocked, so the proportionality does not remain exact.

Why is high rent land less substitutable between uses ?

We attempt to graphically motivate the closing remark of the previous section. In the diagram below, a fixed acreage of land has been arranged along a horizontal spectrum, so that the more fertile acres are grouped at the left, the remainder at the right. The land can be used for crops or sheep; crop acreage is measured from the left; sheep from the right. Two curves indicate the marginal value of an additional acre allocated to each use. We see that crop yields decline fairly steeply as the best land get used up; whilst the placid ruminants care little about land values.



We measure both crop and sheep output in dollars-worth (ie, choose units so that output prices are 1). The margin of cultivation is located at the intersection of the two marginal [value of] output curves. We choose yield functions consistent with the diagram:

$$\text{Crop output} = \text{Crop value} = B_c L_c^{0.5} \quad \text{previously } L_i = A_i X_i^{h(i)} \quad \text{so } h(i) = 2$$

$$\text{Value of additional crop acre} = B_c L_c^{-0.5} / 2$$

$$\text{Rent per Acre} = B_c L_c^{-0.5}$$

$$\text{Sheep output} = \text{Sheep value} = B_s L_s^{0.999} \quad \text{previously } L_i = A_i X_i^{h(i)} \quad \text{so } h(i) = 1$$

$$\text{Value of additional sheep acre} = B_s$$

$$\text{Rent per Acre} = B_s$$

$$\text{Rental ratio} = R = \text{rent per crop acre} / \text{rent per sheep acre} = [B_c / B_s] L_c^{-0.5}$$

$$\text{FOC: } B_c L_c^{-0.5} / 2 = B_s$$

$$\text{or } [B_c / B_s] L_c^{-0.5} = 2$$

$$\text{so } R = 2 \quad \text{ie, } h_{\text{crops}} / h_{\text{sheep}}$$

Intuition: the steeply declining marginal product curve means tight curvature of the crop yield function, leading to low substitutability. But steeply declining marginal product also implies that infra-marginal and hence average yields are much greater than marginal yields, leading to high average rentals.

CRETH assumes (a) smoothly declining marginal yields; and (b) no special pair-wise relations of substitutability. Interesting thought-experiments, which seem to challenge the rent- σ relation, often rely on breaking (a) or (b).